MULTILEVEL COMPUTER MODEL OF WORLD DEVELOPMENT SYSTEM

User Oriented Descriptions

A SERIES: PART V. THE GLOBAL ENERGY SUBMODEL

Günther Fischer

October 1975 WP-75-129

THE GLOBAL ENERGY SUBMODEL

ABSTRACT

Although this model has been presented within the framework of the M.P. Regionalized World Model it is not yet regionalized.

At present the model can be used to study the effects of increasing human energy demand upon natural energy flows. Therefore, the total future energy input from human activity is treated as a scenario variable. As the authors of the model point out in [7] the model, although of sufficient complexity to be useful for understanding energy interactions, is accurate only very near its operating point and is not useful in examining perturbations or stability.

I. MATHEMATICS OF THE MODEL

A. Notation

Since most of the variables and parameters will be explained in Chapter II (concerning data input) only those that are not dealt with in II are reported here.

PSE : Power incident to land used for evaporation.

PLT : Contribution to the thermodynamical cycles from latent heat.

PSS : Power from sunlight contributing to internal energy of the earth, i.e. power available for heating land masses and oceans, and power absorbed in the atmosphere.

PIR : Power lost due to the earth acting as a radiator.

PIT : Internal power contribution to thermodynamical cycles.

PTI : Amount of total power in driving the hydrological and thermodynamical cycles and circulations that re-enter the accumulator of internal energy.

PMI : Power released from winds and currents to internal energy.

PMH : Wind power for human use extracted from the mechanical energy reservoir (including hydropower).

PPC : Gross primary production from power to plants

PPR : Power dissipated from the plant biomass (including respiration, transpiration and any other losses which are proportional to biomass).

PPA : Power contained in plants eaten by herbivores.

PPD : Power contained in plants eaten by decomposers.

PAD : Power dissipated from the animal biomass.

PAH : Animal power to human use.

PPCD : Amount of power allocated to plants that goes to internal energy.

PHIN : Power for producing items useful to man (industry etc.).

PHIND : Depreciation and consumption of energy in EHIN.

PHAD : Power in food consumed.

PHA : Power necessary for producing food.

PHD : Power dissipated in converting power available for man into consumer useful form (e.g. oil into home heat).

PFD : Fossil fuel power dissipated by generating fossil fuels.

PGD : Geothermal power dissipated by extraction.

PND : Nuclear power dissipated by extraction and generation.

RHDND : Ratio of human to natural energy dissipation.

RHDTC : Ratio of human dissipation to energy in weather cycles.

RMDTM : Ratio of human dissipation to energy in thermodynamic cycles.

B. Model Equations

$$PSE = [PSLEK + (1 - PSLRK) \cdot PSILK + PSWEK \cdot (1 - PSWRK) \cdot (1 - PSILK)] \cdot (PSCIK + PSDIK + PSSIK) \cdot PS$$

$$PSS = [(1 - PSLEK - PSLPK) \cdot (1 - PSLRK) \cdot PSILK + (1 - PSWEK - PSWPK) \cdot (1 - PSWRK) \cdot (1 - PSILK)] \cdot (PSCIK + PSDIK + PSSIK) \cdot PS$$

$$PPC = [PSLPK \cdot (1 - PSLRK) \cdot PSILK + PSWPK \cdot (1 - PSWRK) \cdot (1 - PSILK)] \cdot (PSCIK + PSDIK + PSSIK) \cdot PS \cdot PPCK$$

$$PPCD = (1 - PPCK) \cdot PPC/PPCK$$

$$N = 1/SUBDT$$

$$PMH^t = PHMK \cdot PHP^t$$

$$PHIN^t = PHINK \cdot (PPH^t + PAH^t + PHP^t)$$

$$PHA^t = PHAK \cdot (PPH^t + PAH^t + PHP^t)$$

$$PHD^t = PMDK \cdot (PPH^t + PAH^t + PHP^t)$$

$$PFD^t = (1 - PFEK) \cdot PHFK \cdot PHP^t/PFEK$$

$$PGD^t = (1 - PGEK) \cdot PHGK \cdot PHP^t/PGEK$$

$$PND^t = (1 - PNEK) \cdot PHNK \cdot PHP^t/PNEK$$

$$PLT_{l+1}^{t} = PLTK \cdot EL_{l}^{t} \qquad l = 1,\ldots,N-1$$

$$PIR_{l+1}^{t} = PIRK \cdot EI_{l}^{t}$$

$$PIT_{l+1}^{t} = PITK \cdot EI_{l}^{t}$$

$$PTI_{l+1}^{t} = PTIK \cdot (PLTK + PTIK) \cdot EI_{l}^{t}$$

$$PMI_{l+1}^{t} = PMIK \cdot EM_{l}^{t}$$

$$PTM_{l+1}^{t} = PTMK \cdot (PLTK + PITK) \cdot EI_{l}^{t}$$

$$PPR_{l+1}^{t} = PPRK \cdot EP_{l}^{t}$$

$$PPA_{l+1}^{t} = PPAK \cdot EP_{l}^{t}$$

$$PPD_{l+1}^{t} = PPDK \cdot EP_{l}^{t}$$

$$PAD_{l+1}^{t} = PADK \cdot EA_{l}^{t}$$

$$PHIND_{l+1}^{t} = PHINDK \cdot EHIN_{l}^{t}$$

$$PHAD_{l+1}^{t} = PHADK \cdot EHA_{l}^{t}$$

$$EL_{l+1}^{t} = EL_{l}^{t} + (PSE - PLT_{l+1}^{t}) \cdot SUBDT$$

$$\begin{aligned} EI_{l+1}^{t} = EI_{l}^{t} + [&(PSS - PIR_{l+1}^{t} - PIT_{l+1}^{t} + PTI_{l+1}^{t} \\ &+ PMI_{l+1}^{t}) + (PAD_{l+1}^{t} + PPR_{l+1}^{t} + PPCD) \\ &+ (PHD^{t} + PHIND_{l+1}^{t} + PHAD_{l+1}^{t} + PFD_{l+1}^{t} + \\ &PGD_{l+1}^{t} + PND_{l+1}^{t})] \cdot SUBDT \end{aligned}$$

$$EM_{l+1}^{t} = EM_{l}^{t} + (PM_{l+1}^{t} - PMI_{l+1}^{t} - PMH^{t}) \cdot SUBDT$$

$$EP_{l+1}^{t} = EP_{l}^{t} + (PPC - PPR_{l+1}^{t} - PPA_{l+1}^{t} - PPD_{l+1}^{t} - PPH - PPAH) \cdot SUBDT$$

$$EA_{l+1}^{t} = EA_{l}^{t} + (PPAH + PPA_{l+1}^{t} + PPD_{l+1}^{t} - PAD_{l+1}^{t} - PAH) \cdot SUBDT$$

$$EHIN_{l+1}^{t} = EHIN_{l}^{t} + (PHIN^{t} - PHIND_{l+1}^{t}) \cdot SUBDT$$

$$EHA_{l+1}^{t} = EHA_{l}^{t} + (PHA^{t} - PHAD_{l+1}^{t}) \cdot SUBDT$$

$$EL_{1}^{t+1} = EL_{N}^{t}$$

$$EI_{1}^{t+1} = EI_{N}^{t}$$

$$EM_{1}^{t+1} = EM_{N}^{t}$$

$$EP_{1}^{t+1} = EP_{N}^{t}$$

$$EA_{1}^{t+1} = EA_{N}^{t}$$

$$EHIN_{1}^{t+1} = EHIN_{N}^{t}$$

$$EHA_{1}^{t+1} = EHA_{N}^{t}$$

$$PHDTOT^{t} = PFD^{t} + PGD^{t} + PND^{t} + PHD^{t} + PHIND_{N}^{t} + PHAD_{N}^{t}$$

$$PNDTOT^{t} = PSS + PTI_{N}^{t} - PIT_{N}^{t} + PMI_{N}^{t} + PAD_{N}^{t}$$

$$RHDND^{t} = PHDTOT^{t}/PNDTOT^{T}$$

$$RHDTC^{t} = PHDTOT/(PLT_{N}^{t} + PIT_{N}^{t})$$

$$RHDTM^{t} = PHDTOT^{t}/PTM_{N}^{t}$$

II. TERMINAL INPUT AND DATA BASE

A. Requests from the Model

Playing with the model under DOS requires the input of some specific parameters from the keyboard. For this purpose the model issues some appropriate statements on the keyboard. Following the text of each request there is an example for the expected input. This example is primarily intended to show the user the format by which the data are to be entered, rather than to give a meaningful set of data. During a session some or all of the following requests may be issued:

"ENTER SCENARIO NUMBER, E.G. 02"

There are some scenaria which are already implemented and are listed below. If you want to run the model for one of these scenaria you have to type in the corresponding number at this request. If you want to specify parameters yourself you must type "99".

"SCENARIO NUMBER NON-EXISTENT--TRY AGAIN"

This message will be issued if you have typed in a non-existing scenario number. Subsequently you will be asked again for a scenario number.

"ENTER PHP-VALUES, E.G.: 0.011/0.020/0.035/0.050/
0.070/0.090/0.130/0.180/0.280/0.470/0.720/1.100"

At this you must enter a time-series for the power resulting from the estimated human energy demand (12 values). Data points are: 1970; 1980, ..., 2080.

"ENTER PHFK-VALUES, E.G.: 0.540/0.540/0.540/0.540/
0.540/0.540/0.540/0.540/0.540/0.540/0.540/0.540"

This is a time-series with time steps of ten years, which gives the fraction of fossil fuel as part of the total human energy input.

04								NMAX
PMP 01	0.011	0.02	0.035	0.050	0.070	0.090	0.130	
PMP 01	0.180	0.280	0.470	0.720	1.10			
PMFK 01	0.539978	0.539978	0.539978	0.539978	0.539978	0.539978	0.53997	
PMFK 01	0.539978	0.539978	0.539978	0.539978	0.539978			
PMGK 01	0.0099998	0.0099998	0.0099998	0.0099998	0.0099998	0.0099998	0.0099998	
PMGK 01	0.0099998	0.0099998	0.0099998	0.0099998	0.0099998			
PMNK 01	0.299995	0.299995	0.299995	0.299995	0.299995	0.299995	0.299995	
PMNK 01	0.299995	0.299995	0.299995	0.299995	0.299995			
PMP 02	0.011	0.015	0.020	0.027	0.036	0.048	0.065	
PMP 02	0.087	0.117	0.157	0.211	0.284			
PMFK 02	0.539978	0.539978	0.539978	0.539978	0.539978	0.539978	0.53997	
PMFK 02	0.539978	0.539978	0.539978	0.539978	0.539978			
PMGK 02	0.0099998	0.0099998	0.0099998	0.0099998	0.0099998	0.0099998	0.0099998	
PMGK 02	0.0099998	0.0099998	0.0099998	0.0099998	0.0099998			
PMNK 02	0.299995	0.299995	0.299995	0.299995	0.299995	0.299995	0.299995	
PMNK 02	0.299995	0.299995	0.299995	0.299995	0.299995			
PMP 03	0.011	0.016	0.024	0.036	0.053	0.078	0.116	
PMP 03	0.171	0.254	0.375	0.556	0.822			
PMFK 03	0.539978	0.539978	0.539978	0.539978	0.539978	0.539978	0.53997	
PMFK 03	0.539978	0.539978	0.539978	0.539978	0.539978			
PMGK 03	0.0099998	0.0099998	0.0099998	0.0099998	0.0099998	0.0099998	0.0099998	
PMGK 03	0.0099998	0.0099998	0.0099998	0.0099998	0.0099998			
PMNK 03	0.299995	0.299995	0.299995	0.299995	0.299995	0.299995	0.299995	
PMNK 03	0.299995	0.299995	0.299995	0.299995	0.299995			
PMP 04	0.011	0.018	0.029	0.048	0.077	0.126	0.205	
PMP 04	0.335	0.545	0.889	1.445	2.356			
PMFK 04	0.539978	0.539978	0.539978	0.539978	0.539978	0.539978	0.53997	
PMFK 04	0.539978	0.539978	0.539978	0.539978	0.539978			
PMGK 04	0.0099998	0.0099998	0.0099998	0.0099998	0.0099998	0.0099998	0.0099998	
PMGK 04	0.0099998	0.0099998	0.0099998	0.0099998	0.0099998			
PMNK 04	0.299995	0.299995	0.299995	0.299995	0.299995	0.299995	0.299995	
PMNK 04	0.299995	0.299995	0.299995	0.299995	0.299995			

Table 1.

"ENTER PHGK-VALUES, E.G.: 0.010/0.010/0.010/0.010/0.010/
0.010/0.010/0.010/0.010/0.010/0.010/0.0.0"

This time-series gives the fractions of geothermal heat as a part of the total human energy input.

"ENTER PHNK-VALUES, E.G.: 0.300/0.300/0.300/0.300/0.300
0.300/0.300/0.300/0.300/0.300/0.300/0.300"

At last fractions of nuclear power as part of total human energy input are to be entered.

"TO ENTER SCENARIO FROM THE CARD READER HIT < CR > "

You have simply to hit the carriage return key to read in your scenario data. (This will only occur if you have specified scenario number 99.)

The terms appearing in the above requests have the following meaning:

ISCEN : There are scenaria with special sets of parameters available (Table 1). If you want to enter your own parameters you must enter "99". The scenario number is to be entered with format (I2).

PHP : Power resulting from the annual human energy input is computed from the corresponding time-series by use of linear interpolation. The values that you enter at the corresponding request from the model are meant to be the power resulting from total human energy input for the years 1970, 1980, ..., 2080.

PHFK : As already mentioned PHFK is the ratio of fossil fuel to total human energy input. Again it is computed from the corresponding time-series using linear interpolation. Data years are 1970, 1980, ..., 2080

PHGK : Gives the fraction of geothermal heat to total human energy input. Again PHGK is treated the same way as PHFK.

PHNK : Fractions of nuclear energy to total human energy input.

B. Data

In addition to the above time series there are some 47 parameters and initial values necessary to run the model. Under DOS these are read from unit number 2. The data fit on 7 data cards and are read with FORMAT (8F10.7). They are described below:

PS : Solar power (W/m^2).

PPH : Plant power to human use (W/m^2).

PAH : Animal power to human use (W/m^2).

PPAH : Amount of power contained in plants fed to animals which is under man's control (W/m^2).

PPCK : Efficiency of converting wide-spectrum incident sunlight to carbon compounds (excluding albedo loss).

PSCRK : Percentage of incident power reflected into space.

PSCAK : Percentage of incident power absorbed by the droplets.

PSCIK : Percentage of incident power passed to the surface.

PSDAK : Percentage of incident power directly incident to surface.

PSDIK : Percentage of incident power directly absorbed into atmosphere.

PSSRK : Percentage of incident power scattered into space.

PSSIK : Percentage of incident power scattered onto the earth's surface.

PSILK : Percentage of earth's surface that is land.

PSWRK : Average water albedo.

PSWEK : Percentage of power incident to water used for evaporation.

PSWPK : Percentage of power incident to water used for photosynthesis.

PSLRK : Average land albedo.

PSLEK : Percentage of power incident to land masses used for evaporation.

PSLPK : Percentage of power incident to land masses used for photosynthesis.

PHSK : Fraction of total power from human energy input which is solar power.

SUBDT : Factor used in order to stabilize energy flows.

PLTK : Factor related to the power that EL contributes to thermodynamical cycles.

PITK : Factor related to the power that EI contributes to thermodynamical.

PTIK : Factor related to total power in driving hydrological and thermodynamical cycles which re-enters EI.

PTMK : Factor related to power in thermodynamical cycles which drives the winds and currents.

PMIK : Factor related to power released from mechanical energy to internal energy due to viscous dissipation of mechanical energy.

PIRK : Factor related to the power lost due to the earth working as a radiator.

PHMK : Fraction of total power from human energy input which is hydro power.

PPDK : Factor related to power contained in plants eaten by decomposers.

PPAK : Factor related to power contained in plants eaten by herbivores in the natural setting.

PPRK : Factor related to power dissipated by the plant biomass including respiration, transpiration and any other losses that are proportional to biomass.

PADK : Factor related to power dissipated by the animal biomass.

PFEK : Extraction and generation efficiency for fossil fuel.

PGEK : Extraction and generation efficiency for geothermal power.

PNEK : Extraction and generation efficiency for nuclear power.

PHAK : Factor related to power necessary for producing food.

PHINK : Factor related to power for producing items useful to man (in industry, transportation, etc.).

PHDK : Factor related to power dissipated in converting power available for human consumption into consumer useful form (e.g. efficiency of converting oil into home heat).

PHINDK : Factor related to depreciation and consumption of the energy in EHIN.

PHADK : Factor related to power in food consumed and therefore dissipated from EHA.

EL : Reservoir of latent heat of vaporization. Initial value for 1970 (W - yr./m^2).

EM : Reservoir of mechanical energy. Includes winds and currents, and potential and kinetic energy of water on land. Initial value for 1970 (W - yr./m^2).

EI : Reservoir of the internal energy of the earth. Includes sensible heat of the air, land, and water. EI is related to the average global surface temperature of 14^{o} C. Initial value for 1970. (W - yr./m^2)

EP : Energy content of total plant biomass. Initial value for 1970 (W - yr./m^2)

EA : Energy content of the total animal biomass. This includes the top three trophic levels: herbivore, carnivore, top carnivore, and the plant decomposers. 1970 value(W - yr./m^2).

EHIN : Total energy stored in industry, transportation, consumer goods, etc.which was necessary to turn it into items useful to man. Initial value for 1970 (W - yr/m^2)

EHA : Energy in food accumulated for human use. Initial value for 1970. (W - yr/m^2).

As for the use of the model under UNIX all data-including scenario number and time-series for PHP, PHFK, PHGK, and PHNK--are read from unit number 1 which is associated with the filename ENERGY.D

```
01
PHP       0.011      0.02       0.035      0.050      0.070      0.090      0.130
PHP       0.180      0.280      0.470      0.720      1.10
PHFK      0.539978   0.539978   0.539978   0.539978   0.539978   0.539978   0.539978
PHFK      0.539978   0.539978   0.539978   0.539978   0.539978
PHGK      0.0099998  0.0099998  0.0099998  0.0099998  0.0099998  0.0099998  0.0099998
PHGK      0.0099998  0.0099998  0.0099998  0.0099998  0.0099998
PHNK      0.299995.  0.299995   0.299995   0.299995   0.299995   0.299995   0.299995
PHNK      0.299995   0.299995   0.299995   0.299995   0.299995
349.0     0.0014     0.00014    0.0014     1.00
0.249996  0.0099998  0.139996   0.159996   0.259995   0.0699978  0.109997   0.289993
0.079998  0.359993   0.001100   0.079998   0.359993   0.0047998  0.0999985  0.0099998
30.4995   0.259995   0.976974   0.0229993  125.0      4.0        0.0499992  0.0839977
.0139997  0.0999985  0.889984   0.899994   0.899994   0.899994   0.0399990  0.3599930
0.599991  0.1999970  0.289993   1.929990   0.0134995  55.7739    1.79999    0.198994
0.0228992 0.001720
```

Table 2. Data File for use under UNIX

III. OUTPUT

The first page of the printout provides a reproduction of the parameters used to run the model. The next two pages show the results of the computations from 1970 until 2080. Time-series for PHP, RHDND, RHDTC, RHDTM, EI, EL, EM, EP, EA, EHIN, and EHA are given, the most of which has been explained above.

RHDND : Ratio of human to natural energy dissipation.

RHDTC : Ratio of human dissipation to energy in weather cycles.

RHDTM : Ratio of human dissipation to energy in thermodynamic cycles.

The rest of the printout consists of plots for PHP, RHDND, RHDTC, EI, EM, RHDTM, and EHIN.

```
C       WORLD ENERGY CYCLES
        DIMENSION PHPA(12),PHFKA(12),PHGKA(12),PHNKA(12)
        DIMENSION FELD(5,111) ,FELD1(5,111)
        COMMON PS,PPH,PAH,PPAH,PSCRK,PSCAK,PSCIK,PSDAK,PSDIK,PSSRK,PSSIK,
       1PSILK,PSWRK,PSWEK,PSWPK,PSLRK,PSLEK,PSLPK,PHSK,SUBDT,PLTK,PITK,
       2PTIK,PTMK,PMIK,PIRK,PHMK,PPDK,PPAK,PPRK,PADK,PFEK,PGEK,PNEK,PHAK,
       3PHINK,PHDK,PHINDK,PHADK,EL,EH,EI,EP,EA,EHIN,EHA,EM,PPCK
        COMMON EIOLD,EPOLD,EAOLD,EHOLD,EHIOLD
        COMMON PHP,PHFK,PHGK,PHNK,RHDND,RHDTC,RHDTM
        ISTAT = 1970
        MAXIYR = 2080
        NDT = (MAXIYR - ISTAT)/10
        MM = NDT + 1
 30     READ(1,1050) NMAX
        WRITE(6,1040)
        READ(5,1050) ISCEN
        IF(ISCEN .EQ. 99) GO TO 15
        DO 8 JJ = 1,NMAX
        READ(1,1010) (PHPA(J),J = 1,MM)
        READ(1,1010) (PHFKA(J), J = 1,MM)
        READ(1,1010) (PHGKA(J), J = 1,MM)
        READ(1,1010) (PHNKA(J), J = 1,MM)
        IF(ISCEN .EQ. JJ) GO TO 20
     8  CONTINUE
        WRITE(6,1060)
        REWIND 1
        GO TO 30
    15  CONTINUE
        WRITE(6,1150)
        READ(5,1050) IT
        IF(IT.NE.0) GO TO 17
        READ(3,1010) (PHPA(J),J = 1,MM)
        READ(3,1010) (PHFKA(J), J = 1,MM)
        READ(3,1010) (PHGKA(J), J = 1,MM)
        READ(3,1010) (PHNKA(J), J = 1,MM)
        GO TO 20
    17  WRITE(6,1200)
        READ(5,1001) (PHPA(J),J=1,MM)
        WRITE(6,1210)
        READ(5,1001) (PHFKA(J),J=1,MM)
        WRITE(6,1220)
        READ(5,1001) (PHGKA(J),J=1,MM)
        WRITE(6,1230)
        READ(5,1001) (PHNKA(J),J=1,MM)
     20 CONTINUE
        READ(2,1000) PS,PPH,PAH,PPAH,PPCK
        READ(2,1000) PSCRK,PSCAK,PSCIK,PSDAK,PSDIK,PSSRK,PSSIK,PSILK
        READ(2,1000) PSWRK,PSWEK,PSWPK,PSLRK,PSLEK,PSLPK,PHSK,SUBDT
        READ(2,1000) PLTK,PITK,PTIK,PTMK,PMIK,PIRK,PHMK,PPDK
        READ(2,1000) PPAK, PPRK,PADK,PFEK,PGEK,PNEK,PHAK,PHINK
        READ(2,1000) PHDK,PHINDK,PHADK,EL,EM,EI,EP,EA
        READ(2,1000) EHIN,EHA
        WRITE(9,1075)
        WRITE(9,1070)
        DO 1 IT=1,NDT
        IF (IT-1) 2,2,3
     2  CONTINUE
C       INITIALIZE STATE VARIABLES
        GO TO 4
```

```
    3 CONTINUE
C     RESET STATE VARIABLES
      EL = ELOLD
      EM = EMOLD
      EI = EIOLD
      EP = EPOLD
      EA = EAOLD
      EHA= EHAOLD
      EHIN=EHIOLD
    4 CONTINUE
      PHP = PHPA(IT)
      PHFK = PHFKA(IT)
      PHGK = PHGKA(IT)
      PHNK = PHNKA(IT)
      DIFF1 = (PHPA(IT+1) - PHPA(IT))/10.0
      DIFF2 = (PHFKA(IT+1) - PHFKA(IT))/10.0
      DIFF3 = (PHGKA(IT+1) - PHGKA(IT))/10.0
      DIFF4 = (PHNKA(IT+1) - PHNKA(IT))/10.0
      DO 5 J = 1,10
      JJ = J - 1
      PHP = PHPA(IT) + DIFF1*JJ
      PHFK = PHFKA(IT) + DIFF2*JJ
      PHGK = PHGKA(IT) + DIFF3*JJ
      PHNK = PHNKA(IT) + DIFF4*JJ
      CALL ENERGY
      ELOLD = EL
      EMOLD = EM
      EIOLD = EI
      EPOLD = EP
      EAOLD = EA
      EHAOLD = EHA
      EHIOLD =EHIN
      K = (IT - 1)*10 + JJ +1
      KK = ISTAT + K -1
      WRITE(9,1020) KK,PHP,RHDND,RHDTC,RHDTM,EI,EL,EM,EP,EA,EHIN,EHA
      FELD(1,K) = PHP
      FELD(2,K) = RHDND
      FELD(3,K) = RHDTC
      FELD(4,K) = EI
      FELD(5,K) = EM
      FELD1(1,K) = RHDTM
      FELD1(2,K) = EHIN
      FELD1(3,K) = EP
      FELD1(4,K) = EA
      FELD1(5,K) = EL
    5 CONTINUE
    1 CONTINUE
      PHP = PHPA(12)
      PHFK = PHFKA(12)
      PHGK = PHGKA(12)
      PHNK = PHNKA(12)
      CALL ENERGY
      K = MAXIYR - ISTAT + 1
      KK = 2080
      WRITE(9,1020) KK,PHP,RHDND,RHDTC,RHDTM,EI,EL,EM,EP,EA,EHIN,EHA
      FELD(1,K) = PHP
      FELD(2,K) = RHDND
      FELD(3,K) = RHDTC
      FELD(4,K) = EI
```

```
      FELD(5,K) = EM
      FELD1(1,K) = RHDTM
      FELD1(2,K) = EHIN
      FELD1(3,K) = EP
      FELD1(4,K) = EA
      FELD1(5,K) = EL
      DO 100 J = 1,5
      DO 90 JJ = 1,K
      XHELP = FELD(1,JJ)
      FELD(1,JJ) = FELD(J,JJ)
      FELD(J,JJ) = XHELP
   90 CONTINUE
      GO TO (91,92,93,94,95),J
   91 WRITE(9,1080)
      GO TO 96
   92 WRITE(9,1081)
      GO TO 96
   93 WRITE(9,1082)
      GO TO 96
   94 WRITE(9,1083)
      GO TO 96
   95 WRITE(9,1084)
   96 CONTINUE
      CALL BILD(FELD,1,K+1)
      WRITE(9,1090)
  100 CONTINUE
      DO 120 J = 1,2
      DO 110 JJ=1,K
      XHELP = FELD1(1,JJ)
      FELD1(1,JJ) = FELD1(J,JJ)
      FELD1(J,JJ) = XHELP
  110 CONTINUE
      GO TO (111,112,113,114,115),J
  111 WRITE(9,1100)
      GO TO  116
  115 WRITE(9,1110)
      GO TO  116
  113 WRITE(9,1120)
      GO TO  116
  114 WRITE(9,1130)
      GO TO  116
  112 WRITE(9,1140)
  116 CONTINUE
      CALL BILD(FELD1,1,K+1)
      WRITE(9,1090)
  120 CONTINUE
 1000 FORMAT(8F10.7)
 1001 FORMAT(12(F5.3,1X))
 1010 FORMAT(10X,7F10.7)
 1020 FORMAT(1X,I4,2X,F7.4,3(2X,E12.4),2(2X,F7.4),2X,E12.4,3(2X,F7.4),2X
     1,E12.4)
 1040 FORMAT(' ENTER SCENARIO NUMBER, E.G. 02 ',/)
 1050 FORMAT(I2)
 1060 FORMAT(' SCENARIO NUMBER NOT EXISTENT - TRY AGAIN ',/)
 1075 FORMAT(1H1,'M.P. WORLD MODEL : GLOBAL ENERGY CYCLES ',/)
 1070 FORMAT(1H0,'YEAR',4X,'PHP',8X,'RHDND',9X,'RHDTC',9X,'RHDTM',8X,
     1 'EI',7X,'EL',9X,'EM',10X,'EP',7X,'EA',6X,'EHIN',8X,'EHA',/)
 1080 FORMAT(1H1,12X,' PHP = HUMAN ENERGY DEMAND (WATT/M**2)',/)
 1081 FORMAT(1H1,12X,' RHDND = RATIO OF HUMAN TO NATURAL ENERGY DISSIPAT
```

```
     1ION ',/)
 1082 FORMAT(1H1,12X,' RHDTC = RATIO OF HUMAN DISSIPATION TO ENERGY IN W
     1EATHER CYCLES ',/)
 1083 FORMAT(1H1,12X,' EI = EARTH SURFACE INTERNAL ENERGY STORAGE (WATT-
     1YR/M**2) ',/)
 1084 FORMAT(1H1,12X,' EM = EARTH SURFACE MECHANICAL ENERGY STORAGE (WAT
     1T-YR/M**2) ',/)
 1090 FORMAT(12X,'1970',7X,'80',8X,'90',7X,'2000',7X,'10',8X,'20',8X,
     1 '30',8X,'40',7X,'2050',7X,'60',8X,'70',8X,'80',/)
 1100 FORMAT(1H1,12X,'RHDTM = RATIO OF HUMAN DISSIPATION TO ENERGY IN TH
     1ERMODYNAMIC CYCLES ',/)
 1120 FORMAT(1H1,12X,'EP = ENERGY CONTENT OF TOTAL PLANT BIOMASS',/)
 1110 FORMAT(1H1,12X,'EL = ENERGY STORAGE IN LATENT HEAT OF VAPORIZATION
     1 ',/)
 1130 FORMAT(1H1,12X,'EA = ENERGY CONTENT OF TOTAL ANIMAL BIOMASS ',/)
 1140 FORMAT(1H1,12X,'EHIN = ENERGY STORAGE IN INDUSTR., TRANSP., CONSUM
     1ER GOODS ',/)
 1150 FORMAT(1X,' TO ENTER SCENARIO FROM THE CARD READER HIT <CR>',/,
     1' OTHERWISE TYPE 1 AND HIT <CR>',/)
 1200 FORMAT( ' ENTER PHP-VALUES, E.G. 0.011/0.020/0.035/0.050/0.070/'
     1,' 0.090/0.130/0.180/0.280/0.470/0.720/1.100',/)
 1210 FORMAT(' ENTER PHFK-VALUES,E.G. 0.540/0.540/0.540/0.540/0.540/',
     1' 0.540/0.540/0.540/0.540/0.540/0.540/0.540',/)
 1220 FORMAT(' ENTER PHGK-VALUES,E.G. 0.010/0.010/0.010/0.010/0.010/',
     1' 0.010/0.010/0.010/0.010/0.010/0.010/0.010',/)
 1230 FORMAT(' ENTER PHNK-VALUES, E.G. 0.300/0.300/0.300/0.300/0.300/',
     1' 0.300/0.300/0.300/0.300/0.300/0.300/0.300',/)
      STOP
      END
      SUBROUTINE ENERGY
      COMMON PS,PPH,PAH,PPAH,PSCRK,PSCAK,PSCIK,PSDAK,PSDIK,PSSRK,PSSIK,
     1PSILK,PSWRK,PSWEK,PSWPK,PSLRK,PSLEK,PSLPK,PHSK,SUBDT,PLTK,PITK,
     2PTIK,PTMK,PMIK,PIRK,PHMK,PPDK,PPAK,PPRK,PADK,PFEK,PGEK,PNEK,PHAK,
     3PHINK,PHDK,PHINDK,PHADK,EL,EH,EI,EP,EA,EHIN,EHA,EM,PPCK
      COMMON EIOLD,EPOLD,EAOLD,EHOLD,EHIOLD
      COMMON PHP,PHFK,PHGK,PHNK,RHDND,RHDTC,RHDTM
C     ATMOSPHERE POWER FLOW
      PSCR = PSCRK*PS
      PSCA = PSCAK*PS
      PSCI = PSCIK*PS
      PSDA = PSDAK*PS
      PSDI = PSDIK*PS
      PSSR = PSSRK*PS
      PSSI = PSSIK*PS
      PSA  = PSCA+PSDA
      PSI  = PSCI+PSDI+PSSI
      PSIL = PSILK * PSI
      PSIW = (1.0-PSILK)*PSI
C     HYDROSPHERE POWER FLOW
      PSWR=PSWRK*PSIW
      PSW = (1.0-PSWRK)*PSIW
      PSWE= PSWEK*PSW
      PSWP= PSWPK*PSW
      PSWS= (1.0-PSWEK-PSWPK)*PSW
C     GEOSPHERE POWER FLOW
      PSLR = PSLRK*PSIL
      PSL = (1.0-PSLRK)*PSIL
      PSLE=PSLEK*PSL
      PSLP=PSLPK*PSL
```

```
      PSH=PHSK*PHP
      PSLH=PSH
      PSLS= (1.0-PSLEK-PSLPK)*PSL-PSLH
C     PHYSICAL POWER I-O CALCULATIONS
      PSR=PSCR+PSSR+PSWR+PSLR
      PSP=PSLP+PSWP
      PSE=PSLE+PSWE
      PSS=PSLS+PSHS+PSA
C     PHYSICAL ENERGY STORAGE
      NSUB = 1.0/SUBDT+0.1
      DO 10 ISUB = 1,NSUB
      PLT = PLTK*EL
      PIT= PITK*EI
      PTC=PLT+PIT
      PTI=PTIK*PTC
      PTM= PTMK*PTC
      PMI = PMIK*EM
      PIR = PIRK*EI
      PMH = PHMK*PHP
      EL = EL+(PSE-PLT)*SUBDT
      EI = EI+(PSS-PIR-PIT+PTI+PMI)*SUBDT
      EM = EM+(PTM-PMI-PMH)*SUBDT
C     BIOLOGICAL ENERGY FLOW AND STORAGE
      PPC = PPCK*PSP
      PPCD = PSP*(1.0 - PPCK)
C     (PHOTOSYNTHESIS CONV.EFF.INCL. IN PSLPK,PSWPK)
      PPD = PPDK*EP
      PPA = PPAK*EP
      PPR = PPRK*EP
      PAD = PADK*EA
      EP  = EP +(PPC-PPR-PPA-PPD-PPH-PPAH)*SUBDT
      EA =EA + (PPAH+PPA+PPD-PAD-PAH)*SUBDT
      EI = EI + (PAD+PPR+PPCD)*SUBDT
C     CULTURAL ENERGY FLOW AND STORAGE
      PFH = PHFK*PHP
      PF  = PFH/PFEK
      PFD = (1.0-PFEK)*PF
      PGH = PHGK*PHP
      PG  = PGH/PGEK
      PGD = (1.0-PGEK)*PG
      PNH = PHNK*PHP
      PN  = PNH/PNEK
      PND = (1.0-PNEK)*PN
      PH  = PPH + PAH + PHP
C     CULTURAL ENERGY CONSUMPTION
      PHA = PHAK*PH
      PHIN=PHINK*PH
      PHD = PHDK *PH
      PHIND = PHINDK*EHIN
      PHAD = PHADK*EHA
      EHIN = EHIN + (PHIN - PHIND)*SUBDT
      EHA = EHA + (PHA - PHAD) * SUBDT
      EI   = EI + (PHD+PHIND+PHAD+PFD+PGD+PND)*SUBDT
   10 CONTINUE
      PHDTOT =PFD+PGD+PND+PHD+PHIND+PHAD
      PNDTOT =PSS+PTI-PIT+PMI+PAD + PPR
      RHDND = PHDTOT/PNDTOT
      RHDTC = PHDTOT/PTC
      RHDTM = PHDTOT/PTM
```

```
RETURN
END
```

References

[1] Oehmen, K.H., and Paul, W. "Construction of Population Submodels." IIASA Symposium Proceedings SP-74-3, pp. B362-B480.

[2] Clapham, W.B., Shook, T., and Warshaw, M.W. "A Regionalized Food Model for the Global System." IIASA Symposium Proceedings SP-74-3, pp. B500-B554.

[3] Mesarovic, M.D., Richardson, J.M. Jr., Shook, T., and Warshaw, M.W. "The Structural Description and Sensitivity Analysis of the Food Submodel." IIASA Symposium Proceedings SP-74-3, pp. B556-B626.

[4] Gottwald, M., and Pestel, R. "Environmental Impact Assessment." IIASA Symposium Proceedings SP-74-5 pp. B1122-B1276.

[5] Rechenmann, F. "Conversational Use of Multi-Layer Decision Models." IIASA Symposium Proceedings SP-74-6, pp. C232-C246.

[6] Cardenas, M.A., and Huerta, J.M. "Submodel of Global Water Cycle on Regional Basis." IIASA Symposium Proceedings SP-74-5, pp. B1278-B1391.

[7] Heyes, R.P., Jerdonek, R.A., and Kuper, A.B. "Global Energy Submodel." IIASA Symposium Proceedings SP-74-5, pp. B1084-B1119.

[8] Bossel, H. "Energy Supply Submodel." IIASA Symposium Proceedings SP-74-4, pp. B834-B970.